Odhady hustot na základě reálných dat

Pomocné funkce

```
Clear[α, λ, t, f1, f2, EX]
f1 [t_] := α λ (1 - ⅇ^(-t λ))^(α-1) ⅇ^(-t λ)
f2 [t_] := α λ (1 + λ t ) ^ (α - 1) Exp[1 - (1 + λ t ) ^α]

Logf [f1, t_] := Log[α λ (1 - ⅇ^(-t λ))^(α-1) ] - t λ
Logf [f2, t_] := Log[ α λ (1 + λ t ) ^ (α - 1)] + 1 - (1 + λ t ) ^α

EX[K_, f_] := EX[K, f] =
  Assuming[{α > 0, λ > 0}, Integrate[t^K f[t], {t, 0, Infinity}]] // FullSimplify


(*Metoda maximální věrohodnosti*)
MLE[f_ ] := Module[{},
  Clear[α, λ];
  Clear[sectiLogy, ucfc];
  sectiLogy = Table[Logf[f, t], {t, data}];
  ucfc = Total[sectiLogy];
  r = NMaximize[{ucfc, α > 0, λ > 0},
    {{α, 0.1, 20}, {λ, 0.9, 1.2}}, MaxIterations → 100000];
  {α, λ} /. r[[2]]
 ]

(*Momentová metoda*)
MM[f_] :=
 Module[
  {},
  Clear[α, λ];
  mean = Mean[data];
  meanTo2 = Mean[data^2 ];
  Clear[ex1, ex2];
  ex1 [α_, λ_] := EX[1, f];
  ex2[α_, λ_] := EX[2, f];

  res = FindRoot[{ex1[α, λ] == mean, ex2[α, λ] == meanTo2},
    {{α, 5}, {λ, 0.3}}, MaxIterations → 10000];
  {α, λ} /. res
 ]
```

První datová sada

```
data1 =
  WeatherData["KBOS", "TotalPrecipitation", {{2000, 1, 1}, {2005, 12, 31}, "Month"}];
data = data1["Path"][[All, 2]]

r1 = MLE[f1];
r2 = MM[f1];

{α, λ} = r1
f3p = f1 [t] ;
{α, λ} = r2
f2p = f1 [t] ;
```

```
{7.1, 4.4, 8.5, 6.85, 6.54, 16.71, 14.21, 5.87, 7.56, 7.77, 11.55, 10.41, 4.21, 3.31, 18.78,
 2.24, 3.25, 7.11, 5.64, 11.89, 5.86, 2.51, 2.06, 8.31, 6.5, 5.31, 9.21, 7.65, 12.37,
 12.81, 3.79, 5.02, 8.72, 8.58, 13.07, 10.81, 4.18, 6.96, 9.75, 9.76, 10.34, 12.43, 5.11,
 6.82, 6.92, 15.7, 5.39, 9.85, 2.51, 3.71, 5.34, 27.75, 7.87, 4.6, 9.46, 11.24, 15.02,
 5.28, 8.25, 8.48, 8.83, 5.17, 9.06, 7.6, 11.14, 7.45, 8.55, 7.39, 4.65, 24.27, 7.62, 6.62}
```

... FindRoot: The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances.

```
{5.47008, 0.280892}

{4.66872, 0.263254}
```

```
p1 = Plot[{f2p, f3p}, {t, 1, 30},
   PlotRange → All, PlotStyle → {Red, Blue}, PlotTheme → "Scientific",
   PlotLegends → Placed[LineLegend[{Style["Maximálně věrohodný odhad", 10],
       Style["Odhad momentovou metodou", 10]}, LegendFunction → Framed], {{.7, .8}}]];
h1 = Histogram[data, {2}, "PDF", PlotTheme → "Scientific", ChartStyle → Green];
Show[h1, p1]
```



Druhá datová sada

```
data = {12.20, 23.56, 23.74, 25.87, 31.98,
  37, 41.35, 47.38, 55.46, 58.36, 63.47, 68.46, 78.26,
74.47, 81.43, 84, 92, 94, 110, 112, 119, 127, 130, 133, 140, 146,
155, 159, 173, 179, 194, 195, 209, 249, 281, 319, 339, 432, 469,
519, 633, 725, 817, 1776}
```

```
{12.2, 23.56, 23.74, 25.87, 31.98, 37, 41.35, 47.38, 55.46, 58.36, 63.47, 68.46,
 78.26, 74.47, 81.43, 84, 92, 94, 110, 112, 119, 127, 130, 133, 140, 146, 155, 159,
 173, 179, 194, 195, 209, 249, 281, 319, 339, 432, 469, 519, 633, 725, 817, 1776}
```

```
r1 = MLE[f1];
r2 = MLE[f2];

{α, λ} = r1
f2p = f1 [t] ;
{α, λ} = r2
f3p = f2 [t] ;
p1 = Plot[{f2p, f3p}, {t, 20, 2500},
   PlotRange → All, PlotStyle → {Red, Blue}, PlotTheme → "Scientific",
   PlotLegends → Placed[LineLegend[{Style["Maximálně věrohodný odhad pomocí f1", 10],
       Style["Maximálně věrohodný odhad pomocí f2", 10]},
      LegendFunction → Framed], {{.6, .8}}]];
```

```
{1.07137, 0.00468621}
```

```
{0.694028, 0.00855385}
```

```
h1 = Histogram[data, {125}, "PDF", PlotTheme → "Scientific", ChartStyle → Green];
Show[h1, p1]
```



Simulace

```
(*pomocné funkce*)
Clear[α, λ, t, f1, f2, EX]
f1 [t_] := α λ (1 - ⅇ^(-t λ))^(α-1) ⅇ^(-t λ)
f2 [t_] := α λ (1 + λ t ) ^ (α - 1) Exp[1 - (1 + λ t ) ^α]

Logf [f1, t_] := Log[α λ (1 - ⅇ^(-t λ))^(α-1) ] - t λ
Logf [f2, t_] := Log[ α λ (1 + λ t ) ^ (α - 1)] + 1 - (1 + λ t ) ^α

EX[K_, f_] := EX[K, f] =
  Assuming[{α > 0, λ > 0}, Integrate[t^K f[t], {t, 0, Infinity}]] // FullSimplify

(*Metoda maxiální věrohodnosti*)
MLE[𝒟_, f_, N_] := Module[{},
  data = RandomVariate[𝒟, N];
  Clear[α, λ];
  Clear[sectiLogy, ucfc];
  sectiLogy = Table[Logf[f, t], {t, data}];
  ucfc[α_, λ_] := Total[sectiLogy];
  r =
   NMaximize[{ucfc[α, λ], α > 0, λ > 0, α < 0.8, λ < 2}, {α, λ}, MaxIterations → 100000];
  {α, λ} /. r[[2]]
 ]

(*Momentová metoda*)
MM[𝒟_, f_, N_] :=
 Module[
  {},
  data = RandomVariate[𝒟, N];
  Clear[α, λ];
  mean = Mean[data];
  meanTo2 = Mean[data^2 ];
  Clear[ex1, ex2];
  ex1 [α_, λ_] := EX[1, f];
  ex2[α_, λ_] := EX[2, f];

  res = FindRoot[{ex1[α, λ] == mean, ex2[α, λ] == meanTo2},
    {{α, 1}, {λ, 1}}, MaxIterations → 10000];
  {α, λ} /. res
 ]
```

```
(*Momentová metoda speciálně pro zobecnění II pro lepší stabilitu řešení*)
MM2[𝒟_, f_, N_] :=
 Module[
  {},
  data = RandomVariate[𝒟, N];
  Clear[α, λ];
  mean = Mean[data];
  meanTo2 = Mean[data^2];
  Clear[ex1, ex2];
  λ = (-1 + ⅇ Gamma[1 + 1/α, 1]) / mean;
  ex2[α_] := (1 + ⅇ ExpIntegralE[-2/α, 1] - 2 ⅇ Gamma[1 + 1/α, 1]) / ((-1 + ⅇ Gamma[1 + 1/α, 1]) / mean)^2;

  res = FindRoot[ex2[α] == meanTo2, {α, 1}, MaxIterations → 10000]; {
    α, λ} /. res
 ]
```

```
runSimulations[α0_, λ0_, f_, estimatorFunction_] := Module[{},
  α = α0;
  λ = λ0;
  Clear[𝒟];
  𝒟 = ProbabilityDistribution[f[t], {t, 0, Infinity}];
  pocetExperimentu = 10;
  rozsahExporimentu = {100, 1000, 10000};
  SeedRandom[31415926];
  bp = Range[6] 0;
  Table[
   simulationres =
    Table[estimatorFunction[𝒟, f, rozsahExporimentu[[j]]], {i, 1, pocetExperimentu}];
   bp[[j]] = simulationres[[All, 1]];
   bp[[j + 3]] = simulationres[[All, 2]];, {j, 3}];
  bp;

  tbl = Table[{10^(i + 1), Mean [bp[[i]] - α0],
     Variance[bp[[i]]], Mean[(bp[[i]] - α0)^2]}, {i, 3}];
  tbl2 = Table[{Mean [bp[[i]] - λ0], Variance[bp[[i]]],
     Mean[(bp[[i]] - λ0)^2]}, {i, 4, 6}];
  restb = Transpose[Join[Transpose[tbl], Transpose[tbl2]]];
  restb = Round[restb, .00001];
  r1 = TableForm[restb, TableHeadings →
     {None, {"N", "E(α-α)", "Var α", "MSE α", "E(λ-λ)", "Var λ", "MSE λ"}}];


  abp = bp[[1 ;; 3]];
  lbp = bp[[4 ;; 6]];
  {r1, GraphicsRow[{
     BoxWhiskerChart[abp, ChartLabels → {100, 1000, 10000}, PlotTheme → "Scientific",
      AxesLabel → {"Rozsah experimentu", "Odhad parametru"}],
     BoxWhiskerChart[lbp, ChartLabels → {100, 1000, 10000},
      PlotTheme → "Scientific", ChartStyle → Blue]
    }]}
 ]
```

```
simulRes2 = runSimulations[1.5, 1, f1, MLE];
simulRes2[[1]]
simulRes2[[2]]
```

N	E (α−α)	Var α	MSE α	E (λ−λ)	Var λ	MSE λ
100.	0.03627	0.05706	0.05781	0.00837	0.01422	0.01415
1000.	0.01424	0.00379	0.00395	0.00523	0.00125	0.00127
10 000.	−0.00133	0.00039	0.00039	−0.0015	0.00012	0.00012






```
simulRes3 = runSimulations[0.5, 1, f1, MLE];
simulRes3[[1]]
simulRes3[[2]]
```

N	E (α−α)	Var α	MSE α	E (λ−λ)	Var λ	MSE λ
100.	0.02607	0.00511	0.00528	0.03415	0.02011	0.01927
1000.	0.00036	0.00049	0.00044	−0.00855	0.00299	0.00276
10 000.	0.00318	0.00004	0.00004	0.00595	0.00029	0.00029





```
simulRes5 = runSimulations[1.5, 1, f1, MM];
simulRes5[[1]]
simulRes5[[2]]
```

N	E (α−α)	Var α	MSE α	E (λ−λ)	Var λ	MSE λ
100.	0.10622	0.14127	0.15114	0.02865	0.02756	0.0281
1000.	0.02206	0.01243	0.0128	0.00755	0.00274	0.00277
10 000.	−0.00416	0.00128	0.00129	−0.00269	0.00024	0.00024






```
simulRes6 = runSimulations2[0.5, 1, f1, MM];
simulRes6[[1]]
simulRes6[[2]]
```

Interpolation: The point 0.99999997697141` in dimension 1 is duplicated.

Interpolation: The point 0.9999999728094472` in dimension 1 is duplicated.

N	E (α−α)	Var α	MSE α	E (λ−λ)	Var λ	MSE λ
100.	0.04488	0.01622	0.01807	0.06472	0.05971	0.0633
1000.	0.0073	0.00159	0.00163	0.01249	0.0059	0.006
10 000.	−0.00089	0.00018	0.00018	−0.00312	0.00051	0.00052





```
simulRes23 = runSimulations[0.5, 1, f2, MLE];
simulRes23[[1]]
simulRes23[[2]]
```

N	E(α−α)	Var α	MSE α	E(λ−λ)	Var λ	MSE λ
100.	0.02335	0.00814	0.0086	0.00696	0.13383	0.13254
1000.	0.00521	0.00052	0.00054	−0.01442	0.00939	0.00951
10 000.	−0.00071	0.00005	0.00005	0.00262	0.00122	0.00122




```
simulRes23 = runSimulations[0.5, 1, f2, MM2];
simulRes23[[1]]
simulRes23[[2]]
```

N	E(α−α)	Var α	MSE α	E(λ−λ)	Var λ	MSE λ
100.	0.05557	0.01092	0.01291	−0.08375	0.21198	0.19779
1000.	0.00259	0.0007	0.00064	−0.01259	0.01395	0.01271
10 000.	0.00344	0.00013	0.00013	−0.01315	0.00296	0.00283